CHECK WHAT YOU LEARNT

Check Yourself - Unit 3

A. Choose the correct answer

1. A laser differs from an LED mainly because its light is
   (a) brighter (b) coherent and collimated (c) warmer (d) invisible
2. In darkness the resistance of an LDR is
   (a) near zero (b) very high (c) unchanged (d) negative
3. An LDR is unsuitable for high-speed counting because it is
   (a) too small (b) slow to settle (c) too cheap (d) digital
4. Latching is needed because
   (a) the buzzer is quiet (b) a fast event could be missed between loops (c) the LCD is slow (d) the laser flickers
5. An intruder passing with no alarm raised is a
   (a) false positive (b) false negative (c) latch (d) systematic error
6. Real beam sensors resist being fooled by a torch because the beam is
   (a) brighter (b) modulated (c) invisible (d) reflected

B. Fill in the blanks

1 A laser's narrow parallel beam is described as ______________.

2 The light-sensitive track in an LDR is made of ______________ sulphide.

3 The condition && !alarmLatched stops one interruption being counted ______________ times.

4 Ignoring a switch's contact chatter is called ______________.

C. Answer in one or two lines

1 Write down your four measured LDR readings and justify the threshold you chose.

2 Explain why the buzzer is driven with millis() rather than delay().

3 Describe how beam modulation would defeat the torch attack.

4 State your false positive and false negative counts and what they tell you.

UNIT 4

Intermediate Project: Ultrasonic Measurement and Calibration

Sessions 9, 10 and 11

This unit has no glamorous output. It produces a table, a graph and two correction numbers - and it is the most genuinely professional work in the book. Every instrument you will ever trust has been through this process.

By the end of this unit you will be able to:

- Explain how a piezoelectric transducer both emits and detects sound.
- Characterise a sensor against a reference standard across its range.
- Implement mean and median filters and show which is better, with data.
- Derive and apply a gain and offset correction.

SESSION 9

one class

Inside an Ultrasonic Transducer

Today you will

- Learn how a piezoelectric crystal works in both directions.
- Understand beam angle, blind zone and maximum range.
- Mount the sensor for repeatable measurements.

Behind each metal grille on an HC-SR04 is a thin disc of piezoelectric ceramic bonded to a brass plate. **Piezoelectric** means the material physically changes shape when a voltage is applied - and, remarkably, the reverse is also true: squeeze it and it generates a voltage.



A crystal that changes shape

Apply a voltage
the crystal flexes and pushes the air - it is a speaker

Sound hits it
the crystal is squeezed and makes a voltage - it is a microphone

40 kHz is chosen
high enough to be silent to us, low enough to travel usefully far

The two cylinders on an HC-SR04 are the same part used in opposite directions: one pushes air, one listens.

The beam spreads about 15 degrees, so a narrow pole may be missed while a wall behind it is seen instead.

Figure - Inside the HC-SR04: one crystal, two jobs

So the two cylinders are the same component used in opposite directions. Drive one at 40,000 cycles per second and it flexes 40,000 times a second, pushing the air - a speaker. Let the returning sound squeeze the other and it produces a tiny alternating voltage - a microphone. The module's own circuit amplifies that, decides when the echo arrived, and reports the answer as the width of a single pulse on the ECHO pin.

Behind the Scenes

40 kHz is not an arbitrary choice. It must be above roughly 20 kHz so humans cannot hear it. It must not be far above, because higher frequencies are absorbed by air much more strongly and the range would collapse. 40 kHz sits at a comfortable compromise - and it is also why dogs and some insects can hear your sensor perfectly well even though you cannot.

Property	Typical value	What it means for your measurements
Frequency	40 kHz	Silent to us; audible to some animals
Beam angle	about 15 degrees	A narrow pole may be missed and a wall seen behind it
Blind zone	below about 3 cm	The transducer is still ringing; readings are meaningless
Maximum range	about 4 metres	Beyond this the echo is too weak to detect
Settling time	about 60 ms	Measure faster and you may catch the previous echo

Watch Out

The blind zone catches everybody. Place an object 2 cm from the sensor and you will not get 2 cm - you will get nonsense, because the transmitting disc is still vibrating when the echo arrives. Your calibration must therefore start at 10 cm, and your final report must state the working range honestly rather than pretending the sensor works everywhere.

SESSION 10

one class

Characterising Your Own Sensor

Today you will

- Set up a repeatable measurement rig.
- Record readings against a tape across the full range.
- Tabulate error and identify its pattern.

This session produces data, not a demonstration. Fix the sensor at one end of a bench with tape marks at known distances. The target must be a flat, hard surface held square to the beam - a book or a clipboard is ideal, a person's hand is not, because it is neither flat nor repeatable.

Characterise.ino - raw, unfiltered, uncorrected

```
int trig = 9, echo = 10;

void setup() {
  pinMode(trig, OUTPUT);
  pinMode(echo, INPUT);
  Serial.begin(9600);
}

float readOnce() {
  digitalWrite(trig, LOW);
  delayMicroseconds(2);
  digitalWrite(trig, HIGH);
  delayMicroseconds(10);
  digitalWrite(trig, LOW);
  long t = pulseIn(echo, HIGH, 30000);   // 30 ms timeout
  if (t == 0) { return -1; }             // nothing came back
  return t * 0.0343 / 2.0;               // cm, at about 20 degrees C
}

void loop() {
  float d = readOnce();
  Serial.println(d);
  delay(200);
}
```


Note 0.0343 rather than 0.034. The speed of sound in air is about 343 metres per second at 20 degrees Celsius - and it rises by roughly 0.6 metres per second for every degree warmer. On a 35 degree afternoon sound travels about 2.5 percent faster than your formula assumes, which at two metres is an error of five centimetres. Temperature is a genuine source of systematic error in every ultrasonic instrument.

Air temperature	Speed of sound	cm per microsecond	Error at 2 m if you assume 20 C
0 C	331 m/s	0.0331	reads about 3.6 percent short
20 C	343 m/s	0.0343	none - this is the assumption
30 C	349 m/s	0.0349	reads about 1.7 percent long
40 C	355 m/s	0.0355	reads about 3.5 percent long, roughly 7 cm

True distance (cm)	Reading 1	Reading 2	Reading 3	Mean	Error
10	______	______	______	______	______
20	______	______	______	______	______
40	______	______	______	______	______
60	______	______	______	______	______
80	______	______	______	______	______
100	______	______	______	______	______
150	______	______	______	______	______
200	______	______	______	______	______

Try This

Repeat the 60 cm measurement with the target square to the beam, and then tilted at about 30 degrees. A tilted surface reflects the sound away instead of back, and your reading will jump or vanish. Record what happens - it belongs in your limitations section, and it is exactly why a robot can drive into an angled wall it never saw.

SESSION 11

one class

Filtering, Correction and Error Analysis

Today you will

- Plot your data and identify gain and offset error.
- Implement mean and median filters and compare them.
- Apply a correction and prove it worked.

Plot true distance on the horizontal axis and your reading on the vertical. A perfect sensor would give a straight line at 45 degrees passing through the origin. Yours will not - and how it deviates tells you which of two different faults you have.

Characterising your sensor



corrected = raw x gain + offset

gain fixes a reading that drifts more and more with distance; offset fixes a constant shift

Figure 4.1 - Plot it, and the error stops being a mystery

What the plot shows	Name of the fault	The correction
Line is parallel to perfect but shifted up or down	**Offset error**	Subtract a constant

What the plot shows	Name of the fault	The correction
Line starts correct but diverges more and more	**Gain error**	Multiply by a constant
Points scatter widely around the line	**Random error**	Filter - averaging or median
Line bends	**Non-linearity**	Beyond this course - state it as a limitation

Deriving the correction needs only two of your data points, taken far apart - say 20 cm and 150 cm. The **gain** is the ratio of the true span to the measured span. The **offset** is whatever remains after applying the gain.

Deriving gain and offset from two points

```
// Worked example. Suppose at a true 20 cm you read 21.0,
// and at a true 150 cm you read 156.0.
//
//   gain   = (150 - 20) / (156.0 - 21.0)  =  130 / 135.0  =  0.963
//   offset = 20 - (21.0 * 0.963)          =  20 - 20.22   =  -0.22

float gain   = 0.963;      // replace with YOUR numbers
float offset = -0.22;

float corrected(float raw) {
  return raw * gain + offset;
}
```

11.1 Two Filters, and Why the Median Usually Wins

An **averaging filter** adds several readings and divides. It is simple and it does reduce scatter. But it has a weakness: a single wild reading - an echo bounced off a chair leg - drags the average with it.

A **median filter** sorts the readings and takes the middle one. A wild value sorts to the end and is simply ignored. For sensors whose errors are occasional and large - which describes ultrasonic sensors exactly - the median is almost always the better choice.

A five-sample median filter

```
float readMedian() {
  float v[5];
  for (int i = 0; i < 5; i++) {
    v[i] = readOnce();
    delay(60);                      // respect the settling time
  }

  for (int i = 0; i < 4; i++) {      // a simple sort
    for (int j = i + 1; j < 5; j++) {
      if (v[j] < v[i]) {
        float temp = v[i];
        v[i] = v[j];
        v[j] = temp;
      }
    }
  }

  return corrected(v[2]);           // the middle value, then corrected
}
```


At a true 60 cm	Raw	Mean of 5	Median of 5
Highest reading seen	______	______	______
Lowest reading seen	______	______	______

At a true 60 cm	Raw	Mean of 5	Median of 5
Spread, highest minus lowest	______	______	______
Which is tightest?			

> **Challenge**
> Deliberately introduce a wild reading during a run - wave a book across the beam once. Record what the mean does and what the median does. That single experiment is the most convincing paragraph you can put in your report.

New word	What it means
Piezoelectric	Changing shape under voltage, and producing voltage when squeezed.
Beam angle	The cone within which the sensor can see.
Blind zone	The close range where the sensor cannot measure at all.
Gain error	Error that grows in proportion to the reading.
Offset error	A constant shift in every reading.
Median filter	Taking the middle of several sorted readings.