CHECK WHAT YOU LEARNT

Check Yourself - Unit 2

A. Choose the correct answer

1. One microsecond is
   (a) a thousandth of a second (b) a millionth of a second (c) a minute (d) a second
2. pulseIn(echo, HIGH) measures
   (a) voltage (b) how long the pin stays HIGH (c) distance directly (d) temperature
3. A timing value in microseconds should be stored in a
   (a) boolean (b) int (c) long (d) char
4. A function that hands a value back must use
   (a) delay (b) return (c) print (d) loop
5. A motor must not be connected directly to an Arduino pin because
   (a) it is too slow (b) it draws far too much current (c) it is too heavy (d) the pin is analog
6. for (int i = 0; i < 5; i++) runs the block
   (a) 4 times (b) 5 times (c) 6 times (d) forever

B. Fill in the blanks

1 Sound travels about ______________ centimetres in one microsecond.

2 A ______________ is an instruction you write yourself and reuse.

3 The instruction that waits a very short time is delay______________().

4 A motor gets its power from the ______________, not from the Arduino.

C. Answer in one or two lines

1 Explain why milliseconds are too coarse for measuring an echo.

2 Write the first line of a function called readDistance() that returns a whole number.

3 Explain the motor driver using the relay race idea, in your own words.

4 Why might an int variable turn negative when the sensor sees nothing?

UNIT 3

Basic Project: Height Measurement Station

Sessions 6, 7 and 8

Your first machine measures a person's height without a tape and without touching them. It sounds like magic. It is actually one subtraction, wrapped around a sensor that is really a very fast stopwatch.

By the end of this unit you will be able to:

- Explain time-of-flight measurement in your own words.
- Apply a formula inside a program to get a derived quantity.
- Write and call your own readDistance() function.
- Measure your machine's error honestly and reduce it by averaging.

SESSION 6

one class

How an Ultrasonic Sensor Really Works

Today you will

- Understand echo and time of flight.
- Learn the four-step trigger sequence.
- Get your first raw readings onto the Serial Monitor.

The HC-SR04 has two metal cylinders. One is a tiny speaker, the other a tiny microphone. The speaker emits a burst of sound at about 40,000 vibrations per second - far above the roughly 20,000 that human ears can detect, which is why the sensor is called **ultrasonic**. The microphone then listens for that burst to come back.

How an ultrasonic sensor measures without touching



distance = time x 0.034 / 2

time is in microseconds | 0.034 cm is how far sound travels in one microsecond

we divide by 2 because the sound made the journey twice - there and back

Figure 3.1 - Time of flight: the sensor is really a very fast stopwatch

Work through the arithmetic once by hand and it will never confuse you again. Suppose the echo takes **1160 microseconds** to return. Sound covers 0.034 cm per microsecond, so it travelled 1160 x 0.034 = 39.4 cm in total. But that journey was out *and* back, so the object is half that away: **19.7 cm**.

To take a reading you must follow a strict four-step sequence. The 10-microsecond pulse is the sensor's instruction to fire; anything shorter may be ignored.

1. Set TRIG LOW for 2 microseconds, so the line starts clean.
2. Set TRIG HIGH for exactly 10 microseconds. This is the command to fire.
3. Set TRIG LOW again.
4. Call pulseIn(ECHO, HIGH), which waits and reports how long the echo pin stayed HIGH.



Figure 3.2 - Two wires for the sensor, three for the indicator lamps

Sensor pin	Arduino pin	Why
VCC	5V	Power for the sensor
TRIG	D9	Your 10 microsecond command to fire
ECHO	D10	The pin whose HIGH time you measure
GND	GND	Shared return path
Far / medium / near LEDs	D3 / D4 / D5 through 220 ohm	A bar you can read at a glance

Watch Out

TRIG and ECHO are easy to swap, and the symptom is unmistakable: every reading comes back as 0. If that happens, swap the two wires before you touch the code.

SESSION 7

one class

From Distance to Height

Today you will

- Mount the sensor at a known height, facing down.
- Turn a distance reading into a height with one subtraction.
- Write and use your own readDistance() function.

Here is the trick that turns a distance meter into a height meter. Fix the sensor at the top of a stand, pointing straight down, and measure that mounting height once with a tape - call it **H**. When somebody stands underneath, the sensor reports the gap between itself and the top of their head - call that **d**. Their height is simply **H minus d**.

A measuring machine with no tape and no touching



Figure 3.3 - Fix the sensor at a known height and let the maths do the rest

This is a genuinely important idea in engineering, and it has a name: a **derived quantity**. The sensor never measures height at all. It measures a gap, and your program turns that into something a human actually wants to know. Most of the useful numbers in the world are derived this way.

HeightStation.ino

```
int trig = 9;
int echo = 10;
int farLed = 3, midLed = 4, nearLed = 5;

float standHeight = 180.0;   // measure YOUR stand with a tape and change this

void setup() {
  pinMode(trig, OUTPUT);
  pinMode(echo, INPUT);
  pinMode(farLed, OUTPUT);
  pinMode(midLed, OUTPUT);
  pinMode(nearLed, OUTPUT);
  Serial.begin(9600);
}

int readDistance() {              // our own instruction
  digitalWrite(trig, LOW);
  delayMicroseconds(2);
  digitalWrite(trig, HIGH);
  delayMicroseconds(10);         // the command to fire
  digitalWrite(trig, LOW);

  long t = pulseIn(echo, HIGH);  // microseconds the echo pin stayed HIGH
  return t * 0.034 / 2;          // there and back, so halve it
}

void loop() {
  int d = readDistance();
  float height = standHeight - d;

  Serial.print("gap: ");
  Serial.print(d);
  Serial.print(" cm   height: ");
  Serial.print(height);
```

```
  Serial.println(" cm");

  digitalWrite(farLed,  height < 130);   // a short way to write if-else
  digitalWrite(midLed,  height >= 130 && height < 160);
  digitalWrite(nearLed, height >= 160);

  delay(500);
}
```

Look at the three LED lines. `height < 130` is a comparison, and a comparison is already true or false - which is exactly what digitalWrite wants. So the lamp lights when the comparison is true, with no if statement needed at all. It is a neat shortcut worth knowing, though a full if-else is clearer when the logic gets complicated.

SESSION 8

one class

Measuring the Class and Recording Error

Today you will

- Measure classmates and compare against a tape.
- Record your error honestly in a table.
- Reduce noise by averaging several readings.

Now find out how good your machine actually is. Measure five classmates with your station, then measure the same five with a measuring tape. Fill in both columns and work out the difference each time. Do not adjust anything until the table is complete.

Name	Your machine says (cm)	Tape says (cm)	Error (cm)
____________________	____________	____________	____________
____________________	____________	____________	____________
____________________	____________	____________	____________
____________________	____________	____________	____________
____________________	____________	____________	____________
Largest error			____________

You will almost certainly find the readings jump about by a centimetre or two even when nobody moves. That wobble is called **noise**, and every real sensor has it. The professional cure is beautifully simple: take several readings quickly and use their average. One odd reading then counts for only a fifth of the answer instead of all of it.

Averaging to beat the noise

```
int readAverage() {                  // five readings, one steady answer
  long total = 0;
  for (int i = 0; i < 5; i++) {
    total = total + readDistance();
    delay(20);
  }
  return total / 5;
}
```

Watch Out

Hair is a poor reflector of sound - it scatters the echo instead of bouncing it straight back. Thick or curly hair can make a person read several centimetres taller than they are. Place a flat book on the head and your error will drop immediately. Knowing *why* a machine is wrong is worth more marks than pretending it is right.

Challenge

Show the height in feet and inches as well as centimetres. One inch is 2.54 cm. Then use the buzzer to beep once when a stable reading has been held for two seconds, so the person knows they can step away.

New word	What it means
Ultrasonic	Sound too high for human ears to hear.
Time of flight	Measuring distance by timing how long a signal takes to return.
Derived quantity	A useful number worked out from something else that was measured.
Noise	Small random variation in a sensor's readings.
Averaging	Taking several readings and using their mean to reduce noise.